# $Id: PKGBUILD 84285 2013-02-15 14:37:36Z lfleischer $
# Maintainer: Lukas Fleischer <archlinux at cryptocrack dot de>
# Contributor: François Charette <francois ατ archlinux δοτ org>
# Contributor: Giovanni Scafora <giovanni@archlinux.org>

pkgname=jhead
pkgver=2.97
pkgrel=1
pkgdesc='EXIF JPEG info parser and thumbnail remover'
url="http://www.sentex.net/~mwandel/jhead/"
arch=('i686' 'x86_64')
license=('custom')
depends=('glibc')
optdepends=('libjpeg: to up-right images according to rotation tag')
source=("http://www.sentex.net/~mwandel/${pkgname}/${pkgname}-${pkgver}.tar.gz"
        LICENSE)
md5sums=('23b037d0c54211973a3951e41a97c924'
         '1f0db611d2642a981719f45252762a45')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"

  install -Dm755 jhead "${pkgdir}/usr/bin/jhead"
  install -Dm644 jhead.1 "${pkgdir}/usr/share/man/man1/jhead.1"
  install -Dm644 "${srcdir}/LICENSE" "${pkgdir}/usr/share/licenses/${pkgname}/LICENSE"
}
